http://www.mapleprimes.com/questions/200726-Numerical-Integration

```
> restart; interface(version);
  Digits:=15;
  with(IntegrationTools):
  with(plots):

  NumericEventHandler(invalid_operation =
  `Heaviside/EventHandler`(value_at_zero = 1)):
  #Heaviside(0)
```

Classic Worksheet Interface, Maple 17.00, Windows, Feb 21 2013, Build ID 813473

Digits := 15

Task: evaluate the following for $k2 = k1, k3 = -2\,k1$ (as it was already stated)

```
> B3:=-109676400/115648811*(-1/5*cos(b)*sin(w)*sin(b)+k1+k2+k3)*sin(b)*sin(w)*H
  eaviside(-(200*cos(b)*sin(w)*sin(b)-1000*k1-1000*k2-1000*k3)*(-200*cos(b)*sin
  (w)*sin(b)+1000*k1+1000*k2+1000*k3)/abs(-200*cos(b)*sin(w)*sin(b)+1000*k1+100
  0*k2+1000*k3)-117/2)/(1-49/50*cos(b)^2)^(1/2)*cos(b)^2*(abs(-200*cos(b)*sin(w
  )*sin(b)+1000*k1+1000*k2+1000*k3)-117/2)/abs(-200*cos(b)*sin(w)*sin(b)+1000*k
  1+1000*k2+1000*k3);
  ``=subs(k2=k1, k3=-2*k1, %);
  #indets(%, symbol);
```

$$B3 := -\frac{109676400}{115648811}\left(-\frac{1}{5}\cos(b)\sin(w)\sin(b)+k1+k2+k3\right)\sin(b)\sin(w)\,\mathrm{Heaviside}\Big(-(200\cos(b)\sin(w)\sin(b)-1000\,k1-1000\,k2-1000\,k3)\,(-200\cos(b)\sin(w)\sin(b)+1000\,k1+1000\,k2+1000\,k3)\,/\,\left|-200\cos(b)\sin(w)\sin(b)+1000\,k1+1000\,k2+1000\,k3\right|-\frac{117}{2}\Big)\cos(b)^2\left(\left|-200\cos(b)\sin(w)\sin(b)+1000\,k1+1000\,k2+1000\,k3\right|-\frac{117}{2}\right)\Big/\left(\sqrt{1-\frac{49}{50}\cos(b)^2}\left|-200\cos(b)\sin(w)\sin(b)+1000\,k1+1000\,k2+1000\,k3\right|\right)$$

$$=\frac{21935280}{115648811}\cos(b)^3\sin(w)^2\sin(b)^2\,\mathrm{Heaviside}\left(\frac{40000\cos(b)^2\sin(w)^2\sin(b)^2}{\left|-200\cos(b)\sin(w)\sin(b)\right|}-\frac{117}{2}\right)\left(\left|-200\cos(b)\sin(w)\sin(b)\right|-\frac{117}{2}\right)\Big/\left(\sqrt{1-\frac{49}{50}\cos(b)^2}\left|-200\cos(b)\sin(w)\sin(b)\right|\right)$$

```
> k2=k1, k3=-2*k1;
  'Int(Int(Int(B3,a = 0 .. 2*Pi),b = 0 .. 1/2*Pi),w = 0 .. 2*Pi)';
  #  subs(k2=k1, k3=-2*k1, %):

  ``=2*Pi*'Int(Int(B3,b = 0 .. 1/2*Pi),w = 0 .. 2*Pi)';
    subs(k2=k1, k3=-2*k1, %):
  ``=2*Pi*'Int(B3, [b=0 .. Pi/2, w=0 .. 2*Pi], epsilon=1e-4, method=_cuhre)';
    subs(k2=k1, k3=-2*k1, %):
  evalf(%);
```

$$k2 = k1, k3 = -2\,k1$$

$$\int_0^{2\pi}\int_0^{\frac{\pi}{2}}\int_0^{2\pi} B3\,da\,db\,dw$$

$$= 2\pi\int_0^{2\pi}\int_0^{\frac{\pi}{2}} B3\,db\,dw$$

$$= 2\pi\,\mathrm{Int}\left(B3, \left[b = 0\,..\,\frac{\pi}{2}, w = 0\,..\,2\pi\right], \varepsilon = 0.0001, \mathrm{method} = \_cuhre\right)$$

$$= 0.158396643550537$$

re-arranging B3, symmetry in w

```
> '2*Pi*eval( B3, [k2=k1, k3=-2*k1])': '%' = %;
  rhs(%):

  algsubs(200*cos(b)*sin(w)*sin(b) = css, %):
  simplify(%) assuming css::real:
  Bcss:=simplify(%, size)  assuming css::real:

  select(has,%, Heaviside):
  op(%):
  %=simplify(expand(%)) assuming css::real:

  Bcss:=subs(%, Bcss);
  ``;
  '2*Pi*eval( B3, [k2=k1, k3=-2*k1]) = eval(Bcss,
  css=200*cos(b)*sin(w)*sin(b))';
  expand(%): simplify(%) assuming 0<b, b<Pi/2, 0<w,w<2*Pi:
  is(%)  assuming 0<b, b<Pi/2, 0<w,w<2*Pi;
```

$$2\pi\,\mathrm{eval}(B3, [k2 = k1, k3 = -2\,k1]) = \frac{1096764}{578244055}\pi\cos(b)^3\sin(w)^2\sin(b)^2\,\mathrm{Heaviside}\left(\frac{200\cos(b)^2\sin(w)^2\sin(b)^2}{\left|\cos(b)\sin(w)\sin(b)\right|}-\frac{117}{2}\right)\left(200\left|\cos(b)\sin(w)\sin(b)\right|-\frac{117}{2}\right)\Big/\left(\sqrt{1-\frac{49}{50}\cos(b)^2}\left|\cos(b)\sin(w)\sin(b)\right|\right)$$

$$Bcss := \frac{274191}{2891220275}\frac{\left|css\right|\cos(b)\left(\left|css\right|-\frac{117}{2}\right)\pi\,\mathrm{Heaviside}\left(\left|css\right|-\frac{117}{2}\right)}{\sqrt{100-98\cos(b)^2}}$$

$$2\pi\,\mathrm{eval}(B3, [k2 = k1, k3 = -2\,k1]) = Bcss\Big|_{css = 200\cos(b)\sin(w)\sin(b)}$$

true

```
>
> 'Ba = eval( Bcss, css=200*cos(b)*sin(w)*sin(b))';

  convert(Bcss, piecewise, css):
  Ba:= 'eval(%, css=200*cos(b)*sin(w)*sin(b))':
  Ba:= simplify(%, size) assuming 0<b, b<Pi/2, 0<w,w<2*Pi;
  #tmpa:='Ba';

  plot3d(Ba, b=0 .. Pi/2, w=0..2*Pi, axes=boxed, orientation=[10,75,0]);
```

$$Ba = Bcss\Big|_{css = 200\cos(b)\sin(w)\sin(b)}$$

$$Ba := \begin{cases} \dfrac{438705600}{115648811}\dfrac{\pi\left(\frac{117}{400}+\cos(b)\sin(w)\sin(b)\right)\sin(b)\cos(b)^2\sin(w)}{\sqrt{100-98\cos(b)^2}} & 200\cos(b)\sin(w)\sin(b) \le \\ 0 & 200\cos(b)\sin(w)\sin(b) < \\ \dfrac{438705600}{115648811}\dfrac{\pi\sin(b)\left(-\frac{117}{400}+\cos(b)\sin(w)\sin(b)\right)\cos(b)^2\sin(w)}{\sqrt{100-98\cos(b)^2}} & \frac{117}{2} \le 200\cos(b)\sin(w)\,\mathrm{si} \end{cases}$$




```
> 'Ba':
'eval(%, w= Pi - w) - eval(%, w= Pi + w)';
``=convert(%, Heaviside):
combine(%):
expand(%) assuming 0<b,b<Pi/2,0<w,w<Pi:
simplify(%, size);
simplify(%) assuming 0<b,b<Pi/2,0<w,w<Pi;
```

$$Ba\Big|_{w=\pi-w} - Ba\Big|_{w=\pi+w}$$

$$= \frac{256642776}{115648811}\pi\sin(b)\cos(b)^2\sin(w)\left(\mathrm{Heaviside}\left(-50\cos(2b+w)+50\cos(2b-w)+\frac{117}{2}\right)-1\right.$$
$$\left.\mathrm{Heaviside}\left(\frac{117}{2}+50\cos(2b+w)-50\cos(2b-w)\right)-1\right)\Big/\sqrt{100-98\cos(b)^2}$$

$$= 0$$

```
> '2*Int(Ba, [b=0 .. Pi/2, w=0 .. Pi], epsilon=1e-5, method=_cuhre)':
'%'= evalf(%);
```

$$2\,\mathrm{Int}\left(Ba, \left[b=0..\frac{\pi}{2}, w=0..\pi\right], \varepsilon = 0.00001, \mathrm{method} = \_cuhre\right) = 0.158396548560801$$

shift to zero and simplify it

```
> #Bcss: subs(css=200*cos(b)*sin(w)*sin(b), %);
'eval( Bcss, css=200*cos(b)*sin(w)*sin(b))':
'Int(%, [b=0 .. Pi/2, w=0..Pi])';

Change(%, {b=(beta+Pi/4), w=omega+Pi/2}):
subs(beta=b, omega=w, %):

``=map(simplify, %);
rhs(%):
myB:=GetIntegrand(%);
``=Integrand;
```

$$\int_0^{\pi}\int_0^{\frac{\pi}{2}} Bcss\Big|_{css=200\cos(b)\sin(w)\sin(b)}\,db\,dw$$

$$= \int_{-\frac{\pi}{2}}^{\frac{\pi}{2}}\int_{-\frac{\pi}{4}}^{\frac{\pi}{4}} \frac{1096764}{115648811}\cos\left(b+\frac{\pi}{4}\right)\pi\,\mathrm{Heaviside}\left(200\left|\cos\left(b+\frac{\pi}{4}\right)\cos(w)\sin\left(b+\frac{\pi}{4}\right)\right|-\frac{117}{2}\right)$$
$$\left(400\left|\cos\left(b+\frac{\pi}{4}\right)^2\cos(w)^2\sin\left(b+\frac{\pi}{4}\right)^2\right|-117\left|\cos\left(b+\frac{\pi}{4}\right)\cos(w)\sin\left(b+\frac{\pi}{4}\right)\right|\right)\Big/$$
$$\sqrt{100-98\cos\left(b+\frac{\pi}{4}\right)^2}\,db\,dw$$

$$myB := \frac{1096764}{115648811}\cos\left(b+\frac{\pi}{4}\right)\pi\,\mathrm{Heaviside}\left(200\left|\cos\left(b+\frac{\pi}{4}\right)\cos(w)\sin\left(b+\frac{\pi}{4}\right)\right|-\frac{117}{2}\right)$$
$$\left(400\left|\cos\left(b+\frac{\pi}{4}\right)^2\cos(w)^2\sin\left(b+\frac{\pi}{4}\right)^2\right|-117\left|\cos\left(b+\frac{\pi}{4}\right)\cos(w)\sin\left(b+\frac{\pi}{4}\right)\right|\right)\Big/$$
$$\sqrt{100-98\cos\left(b+\frac{\pi}{4}\right)^2}$$

$$= \mathrm{Integrand}$$

That integrand can be simplified: first observe, that within the ranges the funxtion 'absolute' is not needed (plot first to seeit, then prove it):

```
> 'myB';
``=algsubs((cos(b+1/4*Pi)*cos(w)*sin(b+1/4*Pi)) = CCS, %);
CCS=(cos(b+1/4*Pi)*cos(w)*sin(b+1/4*Pi));

plot3d(rhs(%), b=-Pi/4 ... Pi/4, w=-Pi/2...Pi/2, axes=boxed);

0 <= (cos(b+1/4*Pi)*cos(w)*sin(b+1/4*Pi));
is(%) assuming -Pi/4 < b, b < Pi/4, -Pi/2<w,w<Pi/2;
```

myB

$$= \frac{1096764}{115648811} \frac{\pi\,\mathrm{Heaviside}\left(200\,|CCS| - \frac{117}{2}\right) |CCS|\,(400\,|CCS| - 117)\cos\left(b+\frac{\pi}{4}\right)}{\sqrt{100 - 98\cos\left(b+\frac{\pi}{4}\right)^2}}$$

$$CCS = \cos\left(b+\frac{\pi}{4}\right)\cos(w)\sin\left(b+\frac{\pi}{4}\right)$$



$$0 \le \cos\left(b+\frac{\pi}{4}\right)\cos(w)\sin\left(b+\frac{\pi}{4}\right)$$

true

And moreover one can find a simplification for products of the 3 trigonometrics in Heaviside:

```
> CCS:
  %=subs(CCS=(cos(b+1/4*Pi)*cos(w)*sin(b+1/4*Pi)), %); combine(%);
  ``;
  cos(r*b-w)+cos(r*b+w): % = expand(%);         # stupid workaround using
  r=2 ...
  subs(r=2, %);
  %/4;
  CCS=rhs(%);
```

$$CCS = \cos\left(b+\frac{\pi}{4}\right)\cos(w)\sin\left(b+\frac{\pi}{4}\right)$$

$$CCS = \frac{1}{4}\cos(2\,b - w) + \frac{1}{4}\cos(2\,b + w)$$

$$\cos(b\,r - w) + \cos(b\,r + w) = 2\cos(r\,b)\cos(w)$$

$$\cos(2\,b - w) + \cos(2\,b + w) = 2\cos(2\,b)\cos(w)$$

$$\frac{1}{4}\cos(2\,b - w) + \frac{1}{4}\cos(2\,b + w) = \frac{1}{2}\cos(2\,b)\cos(w)$$

$$CCS = \frac{1}{2}\cos(2\,b)\cos(w)$$

```
> CSS = ``;
  (cos(b+1/4*Pi)*cos(w)*sin(b+1/4*Pi)) = 1/2*cos(2*b)*cos(w);
  #plot3d(lhs(%)-rhs(%), b=-Pi/4 ... Pi/4, w=-Pi/2...Pi/2, axes=boxed);
  is(%) assuming -Pi/4 < b, b < Pi/4, -Pi/2<w,w<Pi/2;
```

CSS =

$$\cos\left(b+\frac{\pi}{4}\right)\cos(w)\sin\left(b+\frac{\pi}{4}\right) = \frac{1}{2}\cos(2\,b)\cos(w)$$

true

Thus one can re-write the integrand as follows:

```
> 'myB' = ``; lhs(%):

  algsubs((cos(b+1/4*Pi)*cos(w)*sin(b+1/4*Pi)) = CCS, %):
  subs(abs(CCS) = CCS, %):                 # everything is positive
  algsubs(CCS = cos(2*b)*cos(w)/2, %):    # trigonometric identity from
  above
  theB:=%;
```

myB =

$$theB := \frac{548382}{115648811} \frac{\pi\,\mathrm{Heaviside}\left(\frac{1}{2}\cos(2\,b)\cos(w) - \frac{117}{400}\right)\cos\left(b+\frac{\pi}{4}\right)\cos(2\,b)\cos(w)\,(200\cos(2\,b)\cos(w) - 117)}{\sqrt{100 - 98\cos\left(b+\frac{\pi}{4}\right)^2}}$$

```
> #'theB';
  #plot3d(%, b=-Pi/4 ... Pi/4, w=-Pi/2...Pi/2, axes=boxed);
  '2*Int(theB, [b=-Pi/4 ... Pi/4, w=-Pi/2...Pi/2], epsilon=1e-5,
  method=_cuhre)'; %: evalf(%);
```

$$2\,\mathrm{Int}\left(theB, \left[b = -\frac{\pi}{4} .. \frac{\pi}{4}, w = -\frac{\pi}{2} .. \frac{\pi}{2}\right], \varepsilon = 0.00001, method = \_cuhre\right)$$

0.158396547526113

a different view and Green's theorem

View at the integrand as a product of a function and an indicator

```
> #theB: whattype(%);
  Indicator:=select(has, theB, Heaviside);
  A:=eval(theB, Heaviside=1);
  ``;
  'theB = A * Indicator';
  is(%);
  ``;``;
  Int('A', mu = `D ` .. ``); `D ` = `{(b,w)| Indicator(b,w) = 1)`;
```

$$Indicator := \mathrm{Heaviside}\left(\frac{1}{2}\cos(2\,b)\cos(w) - \frac{117}{400}\right)$$

$$A := \frac{548382}{115648811} \frac{\pi \cos\left(b + \frac{\pi}{4}\right) \cos(2\,b)\cos(w)\,(200\cos(2\,b)\cos(w) - 117)}{\sqrt{100 - 98\cos\left(b+\frac{\pi}{4}\right)^2}}$$

$$theB = A\ Indicator$$

true

$$\int_D A\, d\mu$$

$$D = \{(b,w)|\ Indicator(b,w) = 1\}$$

Then remember that there are formulae relating integration over a domain to integration over its boundary, the most general is Stoke's theorem, Green's theorem is a special case and an online reference is https://de.wikipedia.org/wiki/Satz_von_Green (though a bit sketchy and referencing to a classical book).

```
> F:='F':
  Int('A', mu = `D ` .. ``) = Int('f', sigma = Gamma .. ``),
  Gamma = boundary(`D `);
  ``;``;
  Gamma = `boundary(D) = path of a curve`;
  Int(diff(G(x,y),x) - diff(F(x,y),y), x = `D ` .. ``, y= ``..``) = `curve
  integral using Green's formula`;
```

$$\int_D A\, d\mu = \int_\Gamma f\, d\sigma,\ \Gamma = \mathrm{boundary}(D\ )$$

$$\Gamma = \mathrm{boundary(D) = path\ of\ a\ curve}$$

$$\int\int_D \left(\frac{\partial}{\partial x} G(x,y)\right) - \left(\frac{\partial}{\partial y} F(x,y)\right) dx\, dy = \text{curve integral using Green's formula}$$

From that here we need only a special case:

```
> 'Int( - Diff(F(x,y),y), x = `D ` .. ``, y= ``..``) = Int(F(x,y), x = Gamma
  .. ``)';
  'Int(F(x,y), x = Gamma .. ``) = Int( F(phi(t),phi(t)) * Diff(phi(t), t), t
  = 0 .. 1)';
```

$$\int\int_D -\left(\frac{\partial}{\partial y} F(x,y)\right) dx\, dy = \int_\Gamma F(x,y)\, dx$$

$$\int_\Gamma F(x,y)\, dx = \int_0^1 F(\phi(t), \phi(t)) \left(\frac{d}{dt}\phi(t)\right) dt$$

Here ϕ is a (piecewise differentiable) parametrization of the boundary Γ and it has to have positive orientation (counterclock wise), a closed curve.

```
> 'Indicator'; #Heaviside(1/2*cos(2*b)*cos(w)-117/400);
  plot3d(%, b=-Pi/4 ... Pi/4, w=-Pi/2...Pi/2, axes=boxed,
  scaling=constrained,
    orientation=[-127,54]); #, numpoints=100^1, symbol=point);
  #plots[implicitplot](%, b=-Pi/4 ... Pi/4, w=-Pi/2...Pi/2);

  Heaviside(1/2*cos(2*b)*cos(w)-117/400);
  op(%):
  theCurve:=%*2;
  plots[implicitplot](%, b=-Pi/4 ... Pi/4, w=-Pi/2...Pi/2,
  scaling=constrained, color=red);
```

$$Indicator$$



$$\mathrm{Heaviside}\left(\frac{1}{2}\cos(2\,b)\cos(w) - \frac{117}{400}\right)$$

$$theCurve := \cos(2\,b)\cos(w) - \frac{117}{200}$$



That looks like an ellipse, but it is not an ellipse.

parameterizing the curve (new)

First determine the extrema w.r.t. to the x-axis and y-axis (it would not have been necessary to shift into 0, but now it makes things more easy):

```
> bmax:='bmax':
  0=eval(theCurve, b=bmax): eval(%, w=`0`); eval(%,`0`=0): bmax:=solve(%,
  bmax);
  ``;
  wmax:='wmax':
  0=eval(theCurve, w=wmax): eval(%, b=`0`/2); eval(%,`0`=0): wmax:=solve(%,
  wmax);
  ``;
  '[bmax,wmax]': '%'= evalf(%);
```

$$0 = \cos(2\,\mathrm{bmax})\cos(0) - \frac{117}{200}$$

$$\mathrm{bmax} := \frac{1}{2}\arccos\left(\frac{117}{200}\right)$$

$$0 = \cos(0)\cos(\mathrm{wmax}) - \frac{117}{200}$$

$$\mathrm{wmax} := \arccos\left(\frac{117}{200}\right)$$

$$[\mathrm{bmax}, \mathrm{wmax}] = [0.472958134516794, 0.945916269033588]$$

It can be seen, that th curve is given by a function of b, for the upper and for the lower half plane

```
> theCurve;
  h:= b -> arccos(117/200*1/cos(2*b));
  ``;
  '[h(b),-h(b)]';
  plot(%, b=-bmax..bmax,w=-wmax..wmax, scaling = constrained,
  color=[magenta,blue]);
```

$$\cos(2\,b)\cos(w) - \frac{117}{200}$$

$$h := b \rightarrow \arccos\left(\frac{117}{200}\,\frac{1}{\cos(2\,b)}\right)$$

$$[h(b), -h(b)]$$



Now for b we start in the minimum = $-\mathrm{bmax}$ and run to bmax along the blue line given by $-h(b)$ where we run back to the minimum along the pink line using $h(b)$. That gives a closed curve with positive orientation. As formula one can write it down as follows:

```
> 'X(t)' = 'piecewise(t <= 1/2,4*bmax*t-bmax,1/2 < t,-4*bmax*t+3*bmax)';
  'Y(t)' = 'piecewise(t <= 1/2,-h(X(t)),1/2 < t,h(X(t)))';
  ``;
  piecewise(t <= 1/2,4*bmax*t-bmax,1/2 < t,-4*bmax*t+3*bmax):
  X:=unapply(%, t);
  piecewise(t <= 1/2,-h(X(t)),1/2 < t,h(X(t))):  simplify(%):
  Y:=unapply(%, t);
```

$$X(t) = \begin{cases} 4\,\mathrm{bmax}\,t - \mathrm{bmax} & t \le \frac{1}{2} \\ -4\,\mathrm{bmax}\,t + 3\,\mathrm{bmax} & \frac{1}{2} < t \end{cases}$$

$$Y(t) = \begin{cases} -h(X(t)) & t \le \frac{1}{2} \\ h(X(t)) & \frac{1}{2} < t \end{cases}$$

$$X := t \to \mathrm{piecewise}\left(t \le \frac{1}{2},\ 2\arccos\left(\frac{117}{200}\right) t - \frac{1}{2}\arccos\left(\frac{117}{200}\right),\ \frac{1}{2} < t,\ -2\arccos\left(\frac{117}{200}\right) t + \frac{3}{2}\arccos\left(\frac{117}{200}\right)\right)$$

$$Y := t \to \mathrm{piecewise}\left(t \le \frac{1}{2},\ -\arccos\left(\frac{117}{200}\,\frac{1}{\cos\left(\arccos\left(\frac{117}{200}\right)(4t-1)\right)}\right),\ \frac{1}{2} < t,\ \arccos\left(\frac{117}{200}\,\frac{1}{\cos\left(\arccos\left(\frac{117}{200}\right)(4t-3)\right)}\right)\right)$$

For me that was an ugly task, even if it now reads quite simple. Sigh. Check the orientation by an animated plot (click to play it);

```
> phi:= X +Y*I;
  animate( complexplot, [evalf('phi'(t)), t=0..tau, -1 ..1, -1 .. 1 , color
  = red], tau=0.01 .. 1, frames=15);
```

$$\phi := X + Y\,I$$

tau = .1e-1



I do not actually prove all the 'topological' stuff involved (more by visualization), but at least show, that this parametrization vanishes on the curve

```
> 'eval(theCurve, [b=X(t), w=Y(t)])':
  '%'= simplify(%);
```

$$\mathrm{eval}(\mathrm{theCurve}, [b = X(t), w = Y(t)]) = 0$$

```
> # inner points
```

```
#Heaviside(theCurve); subs(b=1/4,w=1/4, %); simplify(%);
```

using Green's theorem for the final computation

To apply Greens' theorem we need some 'good' function F such that $A = \frac{\partial}{\partial w} F(b, w)$.

Concurrent Maple does it by using a brute anti-derivative (inspect in case of doubts, starting with a plot would a reasonable idea).

```
> 'F(b,w)' = 'int(A, w)';
  rhs(%):
  F:=unapply(%, b,w);

  #plot3d(F(b,w), b=-Pi/4 .. Pi/4, w=-Pi/2 .. Pi/2, axes=boxed);

  'A=diff(F(b,w),w)';
  is(%);
```

$$F(b, w) = \int A\,dw$$

$$F := (b, w) \to \frac{548382}{115648811}\,\frac{\pi \cos\left(b + \frac{\pi}{4}\right)\cos(2b)\left(200\cos(2b)\left(\frac{1}{2}\cos(w)\sin(w) + \frac{1}{2}w\right) - 117\sin(w)\right)}{\sqrt{100 - 98\cos\left(b + \frac{\pi}{4}\right)^2}}$$

$$A = \frac{\partial}{\partial w} F(b, w)$$

$$\mathrm{true}$$

To sum up we now have the following:

```
> '2*Int(theB, [b=-Pi/4 ... Pi/4, w=-Pi/2...Pi/2])' =
  '2*Int(-F(X(t), Y(t)) * diff(X(t),t), t=0..1)';
```

$$2\int_{-\frac{\pi}{2}}^{\frac{\pi}{2}}\int_{-\frac{\pi}{4}}^{\frac{\pi}{4}} \mathrm{theB}\,db\,dw = 2\int_0^1 -F(X(t), Y(t))\left(\frac{d}{dt}X(t)\right)dt$$

Now remember the picture of the curve and that the parametrization is 'piecewise', so in t=1/2 it may not be differentiable. But we have had symmetry along the x-axis

```
> 'Int(-F(X(t), Y(t)) * diff(X(t),t), t=0..1)';
  plot(op(%));

  '-F(X(t), Y(t)) * diff(X(t),t)':
  'eval(%, t = 1/2 +t)=eval(%, t = 1/2 -t)'; %:
  simplify(%) assuming 0<t,t<1/2:
  is(%);
```

$$\int_0^1 -F(X(t), Y(t))\left(\frac{d}{dt}X(t)\right)dt$$



$$\left(-F(X(t), Y(t)) \left(\frac{d}{dt} X(t)\right)\right)\Big|_{t = 1/2 + t} = \left(-F(X(t), Y(t)) \left(\frac{d}{dt} X(t)\right)\right)\Big|_{t = 1/2 - t}$$

$$\text{true}$$

So - finally - we arrive at the following, which can be evaluated without much pain:

```
> 'Int(Int(Int(B3,a = 0 .. 2*Pi),b = 0 .. 1/2*Pi),w = 0 .. 2*Pi)':
  'eval(%, [k2=k1])':
  'eval(%, [k3=-2*k1])';
  ``= '2*Int(theB, [b=-Pi/4 ... Pi/4, w=-Pi/2...Pi/2])';
  ``= - 2*'2*Int(F(X(t), Y(t))* diff(X(t),t), t=0..1/2, method=_d01ajc)';
  #convert(%, piecewise, t) assuming 0<t, t<1/2;
  evalf(%);
```

$$\int_0^{2\pi} \int_0^{\frac{\pi}{2}} \int_0^{2\pi} \mathrm{B3}\, da\, db\, dw \Big|_{k2 = k1} \Big|_{k3 = -2\, k1}$$

$$= 2 \int_{-\frac{\pi}{2}}^{\frac{\pi}{2}} \int_{-\frac{\pi}{4}}^{\frac{\pi}{4}} \mathrm{theB}\, db\, dw$$

$$= -4\, \mathrm{Int}\left(F(X(t), Y(t)) \left(\frac{d}{dt} X(t)\right), t = 0 .. \frac{1}{2}, \mathrm{method} = \_d01ajc\right)$$

$$= 0.158396539759226$$

In case of need for higher precision one can use the formula as follows

```
> 'F(X(t), Y(t))* diff(X(t),t)':
  simplify(%) assuming 0<t,t<1/2:
  newtask:=unapply(%, t):
  -4*Int(newtask, 0 .. 1/2, method = _Dexp);
  ``=evalf[60](%);
```

$$-4\, \mathrm{Int}\left(\mathrm{newtask}, 0 .. \frac{1}{2}, \mathrm{method} = \_Dexp\right)$$

$$= 0.158396539759225856198591427996201908864896997323188018076446$$